AST4310 - Obligatory Exercise 2
Stellar Spectra B: LTE Line Formation

- The pdf of the exercise and supporting material can be found through the AST4310 course website.
- The deadline for handing in the report is: ~~**Friday 16 November 2018**~~ **Friday 23 November 2018** (the original deadline is moved a week later).
- The exercise consists of 3 Chapters. For the report, you should focus on the presentation of Chapter 2 "Continuous spectrum of the solar atmosphere". For this Chapter, you should write a full report with detailed discussion and interpretation of what you show in the figures. For the other two Chapters, 1 and 3, you may restrict to only showing figures with extensive captions that discuss the content. Check the feedback you received on SSA and show that you take this into account.
- The most challenging part of the exercise is in Chapter 3, the LTE formation of the Na D lines in the solar atmosphere.
- Note the units of the various parameters that are studied throughout the exercise. Units are a notorious source of error!
- On page 23, a formula is given as approximation for the partition function of Na I. The logarithms in this formula are in base 10. So in IDL:

```
function parfunc_Na, temp
; partition functions Na
; input: temp (K)
; output: fltarr(3) = partition functions U1,U2,U3
 u=fltarr(3)
 theta=5040./temp
; partition function Na I : Appendix D of Gray (1992)
 c0=0.30955
 c1=-0.17778
 c2=1.10594
 c3=-2.42847
 c4=1.70721
 logU1=c0 + c1 * alog10(theta) + c2 * alog10(theta)^2 + c3 * alog10(theta)^3 $
   + c4 * alog10(theta)^4
 u[0]=10^logU1
; partition function Na II and Na III: approximate by the statistical weights of
; the ion ground states
 u[1]=1  ; from Allen 1976
 u[2]=6  ; from Allen 1976
 return,u
end
```

- λ_0 in the Dopplerwidth $\Delta\lambda_D$ (22) is the center wavelength of the Na D lines.
- There have been changes to the IDL `VOIGT(a,v)` function compared to when SSB was developed. It is normalised to $\sqrt{\pi}$. The correct way to call it is

```
 voigt_NaD =  voigt(a_voigt, v_voigt) / dopplerwidth
```

  where `a_voigt` is given by (21), `v_voigt` by (20), and `dopplerwidth` by (22). `voigt_NaD` is now $H(a,v)/\Delta\lambda_D$ in (14).

- Note the electron charge in (14) in cgs: $e = 4.803204 \times 10^{-10}$ statcoulomb.
- You should include the gas pressure to your call to the function that computes the Na D extinction: `NaD_ext(wav, s, temp, nel, nhyd, vmicro, pgas)`, which is generalized to include both Na D_1 (s=2) and Na D_2 (s=3).
- For Van der Waals broadening, we use the Unsöld recipe (25). Here, the logarithms are in base 10. So in IDL:

```
function gammavdw_NaD, temp, pgas, s
;
; Van der Waals broadening for Na D1 and Na D2
; s=2 : Na D1
; s=3 : Na D2
; using classical recipe by Unsold
; following recipe in SSB
 rsq_u=rsq_NaD(s)
 rsq_l=rsq_NaD(1) ; lower level D1 and D2 lines is ground state s=1
 loggvdw=6.33 + 0.4*alog10(rsq_u - rsq_l) + alog10(pgas) - 0.7 * alog10(temp)
 return, 10.^loggvdw
end

function rsq_NaD, s
; compute mean square radius of level s of Na D1 and Na D2 transitions
; s=1 : ground state, angular momentum l=0
; s=2 : Na D1 upper level l=1
; s=3 : Na D2 upper level l=1
  h=6.62607D-27                  ; Planck constant (erg s)
  c=2.99792D10                   ; light speed [cm/s]
  erg2eV=1/1.60219D-12           ; erg to eV conversion
  E_ionization = 5.139           ; [eV] ionization energy
  E_n=fltarr(3)                  ; energy level: E_n[0]=0 : ground state
  E_n[1]=h*c/5895.94D-8 * erg2eV ; Na D1: 2.10285 eV
  E_n[2]=h*c/5889.97D-8 * erg2eV ; Na D2: 2.10498 eV
  Z=1.                           ; ionization stage, neutral Na: Na I
  Rydberg=13.6                   ; [eV] Rydberg constant
  l=[0.,1.,1.]                   ; angular quantum number
  nstar_sq = Rydberg * Z^2 / (E_ionization - E_n[s-1])
  rsq=nstar_sq / 2. / Z^2 * (5*nstar_sq + 1 - 3*l[s-1]*(l[s-1] + 1))
  return, rsq
end
```

- It is a good idea to check that the various IDL functions give reasonable output. See Figure 1.
- The Na D1 line profile computed with the different recipes in SSB should look like Figure 2.



Figure 1: *Left:* Boltzmann distribution for 3 levels in FALC. *Right:* Saha distribution for Na in FALC.



Figure 2: The Na D1 LTE line profile in FALC using the Unsöld recipe for Van der Waals broadening (solid line). The dashed line is from the FTS atlas.